

## Práctica 9. Análisis de respuesta transitoria. Sistemas de primer orden

**Objetivo.** Obtener y estudiar con ayuda de Scilab y Xcos las características más importantes de la respuesta de sistemas lineales ante entradas típicas, tales como impulso y escalón. Analizar y caracterizar el caso de sistemas de primer orden.

### Introducción.

El modelado de sistemas mediante función de transferencia está basado en la idea de que un sistema se puede caracterizar en términos externos, es decir, en base a experimentos Estimulo - Respuesta, es decir, experimentos Entrada - Salida, ver figura 9.1.



Figura 9.1.- Experimento entrada-salida

Por esta razón, los experimentos consistentes en introducir una señal de prueba a la entrada de un sistema para observar su respuesta (el comportamiento de la salida que produce) son fundamentales para caracterizar lo que podemos esperar de un sistema dinámico.

### Entradas de prueba típicas

Dependiendo del aspecto que nos interesa estudiar de un sistema, se puede utilizar alguna de las siguientes entradas de prueba:

- **Impulso unitario (Delta de Dirac  $\delta(t)$ ).**.- Señal teórica de área unitaria, que vale cero en todos lados excepto en  $t=0$ , tiene duración infinitamente pequeña pero amplitud infinitamente grande. Es usada para dos objetivos:
  - Para obtener la **respuesta sin entrada a condiciones iniciales no nulas**: La respuesta  $y(t)$  del sistema sin entrada ante la condición inicial  $y^{(n-1)}(0) = y_0$  es proporcional a la respuesta del sistema con condiciones iniciales cero al impulso no unitario  $y_0 \delta(t)$ .
  - Como **modelo matemático** del sistema que permite obtener la respuesta del sistema ante cualesquier otra entrada. Si  $h(t)$  es la respuesta del sistema al impulso unitario, entonces la respuesta  $y(t)$  del sistema ante otra entrada arbitraria  $x(t)$  está dada por la **integral de convolución** siguiente

$$y(t) = \int_0^t x(\tau)h(t-\tau)d\tau \quad (9.1)$$

- **Escalón unitario ( $u(t)$ ).** Señal de amplitud constante de valor 1 para  $t \geq 0$ , vale cero para tiempos negativos, es decir,

$$u(t) = \begin{cases} 1 & t \geq 0 \\ 0 & t < 0 \end{cases} \quad (9.2)$$

La señal escalón es utilizada para estudiar la respuesta de un sistema ante entradas constantes, especialmente el comportamiento del error en estado estable.

- **Señal rampa unitaria ( $r(t)$ ).** Señal que se incrementa a partir de  $t=0$  en valores iguales al tiempo transcurrido (es decir, con pendiente unitaria), es decir,

$$r(t) = \begin{cases} t & t \geq 0 \\ 0 & t < 0 \end{cases} \quad (9.3)$$

La señal rampa es utilizada para estudiar la respuesta de un sistema ante entradas que crecen con velocidad constante, especialmente el comportamiento del error en estado estable ante este tipo de entradas.

En la figura 9.2 se muestra la gráfica respecto al tiempo de las señales de prueba típicas anteriores.




Figura 9.2.- Señales de prueba típicas.

### El Comando **csim** de Scilab para simular un sistema lineal.

En la práctica No. 5 se explicó cómo se puede utilizar el comando **ode** para obtener la respuesta de un sistema lineal o no lineal ante entradas o condiciones iniciales arbitrarias.

Aunque siempre es posible utilizar **ode**, Scilab proporciona la función **csim**, la cual es más sencilla y especializada en sistemas lineales definidos por su función de transferencia o por su modelo en espacio de estado mediante **syslin**. La instrucción **csim** se puede ejecutar mediante alguna de las siguientes dos sintaxis:

```
[y] = csim(u,t,G[,tol])
[y [,x]] = csim(u,t,Sist,[x0 [,tol]])
```

Donde:

- **G** representa un sistema definido por su función de transferencia mediante **syslin**.
- **Sist** representa un sistema definido por su modelo de espacio de estado mediante **syslin**.
- **y** es la matriz cuyos renglones contienen los valores de las salidas del sistema.
- **t** es el vector renglón de instantes de tiempo en los que se calcula la respuesta.

- **x** es la matriz cuyos renglones contienen los valores de los estados del sistema.
- **x0** es el vector (columna) de valores iniciales del vector de estado. Si no se especifican las condiciones iniciales **x0**, se suponen cero.
- **tol** es un vector [**atol rtol**] definiendo la tolerancia absoluta y relativa para el *solver*.
- **u** es la entrada al sistema y puede ser especificada de diversas maneras:
  - **u** puede ser una función **entrada=u(t)**
  - **u** puede ser un vector conteniendo los valores de la entrada en cada instante de tiempo t.
  - **u** puede ser una lista: **list (ut,parametro\_1,...,parametro\_n)** tal que: **entrada = ut(t,parametro\_1,...,parametro\_n)** (**ut** es una función)
  - **u** puede ser una cadena (Para sistemas de una sola entrada):
    - La cadena "**impuls**" especifica una entrada tipo impulso unitario, en este caso, las condiciones iniciales se suponen cero. (Para sistemas de grado relativo cero, el impulso de la respuesta en t=0 es ignorado)
    - La cadena "**step**" especifica una entrada tipo escalón unitario, en este caso también se suponen condiciones iniciales cero.

☞ Los términos entre corchetes son opcionales.

**Ejemplo.-** Para obtener la respuesta al escalón unitario del sistema cuya función de transferencia es

$$G(s) = \frac{s+4}{s^4 + 2s^3 + 4s^2 + 2s + 2}$$

Con los comandos de Scilab siguientes se obtiene la respuesta mostrada en la figura 9.3.

```
num=poly([4 1],'s','c');
den=poly([2 2 4 2 1],'s','c');
G=syslin('c',num,den);
t=0:0.1:50;
y=csim('step',t,G);
plot(t',y');
title('Ejemplo de respuesta al escalón unitario')
xgrid
```




Figura 9.3.- Respuesta al escalón unitario de un sistema descrito en función de transferencia.

**Ejemplo.** - Para obtener la respuesta al escalón unitario del siguiente sistema descrito en variables de estado.

$$\dot{x} = \begin{bmatrix} -2 & -4 & -2 & -2 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \end{bmatrix} x + \begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \end{bmatrix} u$$

$$y = [0 \ 0 \ 1 \ 4]x$$

Las instrucciones en Scilab son las siguientes. En la figura 9.4 se muestra la respuesta del sistema, incluyendo no solo la salida, sino también los estados del sistema.

```
A=[-2 -4 -2 -2;1 0 0 0;0 1 0 0;0 0 1 0];
B=[1;0;0;0];
C=[0 0 1 4];
D=[0];
Sist=syslin('c',A,B,C,D);
[y,x]=csim('step',t,Sist);
plot(t',y',t',x');
xlabel('Otro ejemplo de respuesta al escalón unitario'); xgrid;
legend('Salida', 'estado x1', 'estado x2', 'estado x3', 'estado x4');
```




Figura 9.4.- Respuesta al escalón unitario de un sistema descrito en variables de estado

### Relación entre la respuesta ante diferentes entradas de prueba.

Para sistemas lineales es posible obtener la respuesta de un sistema ante una entrada de prueba a partir de su respuesta a otra, observando que existe una relación muy simple entre las entradas de prueba consideradas:

☞ La rampa es la integral del escalón, el cual es la integral del impulso, es decir:

$$r(t) = \int_0^t u(\tau) d\tau, \quad u(t) = \int_0^t \delta(\tau) d\tau \quad (9.4)$$

o lo que es lo mismo:

☞ El impulso es la derivada del escalón, el cual es la derivada de la rampa:

$$\delta(t) = \frac{du(t)}{dt}, \quad u(t) = \frac{dr(t)}{dt} \quad (9.5)$$

Como en sistemas lineales el orden de las operaciones se puede intercambiar, entonces, para un sistema dado por su función de transferencia  $G(s)$ :

☞ La respuesta al impulso unitario de  $G(s)$  es la respuesta al escalón unitario de  $sG(s)$

☞ La respuesta a la rampa unitaria de  $G(s)$  es la respuesta al escalón unitario de  $\frac{G(s)}{s}$

☞ La respuesta a la rampa unitaria de  $G(s)$  es la respuesta al impulso unitario de  $\frac{G(s)}{s^2}$

☞ La respuesta al escalón unitario de  $G(s)$  es la respuesta al impulso unitario de  $\frac{G(s)}{s}$

☞ etc...

**Ejemplo:** Obtener la respuesta a la rampa unitaria por diferentes métodos para el sistema dado por la siguiente función de transferencia

$$G(s) = \frac{1}{s^2 + s + 1}$$

**Primer método.: Generando una función rampa y usando el comando `csim`:**

```
s=poly([0 1], 's', 'c'); //Variable de Laplace
G=1/(s^2+s+1); //función de transferencia del sistema
dt=0.2; //incremento entre instantes de tiempo
tfin=8; //Instante final del tiempo de simulación
t=0:dt:tfin; //Genera el intervalo de tiempo de la entrada
// Problema: ¿cómo saber cuál es el valor más adecuado de tf?
r=t; //Genera la entrada rampa unitaria
y1=csim(r,t,G); //Obtiene la respuesta
plot(t,y1,'b+'); //Grafica respuesta en azul con marcas '+'
plot(t,r,'k--'); // Grafica la rampa en negro con guiones.
xgrid;
```

**Segundo método. Usando la convolución con la respuesta al impulso unitario (9.1):**

```
s=poly([0 1], 's', 'c'); //Variable de Laplace
G=1/(s^2+s+1); //función de transferencia del sistema
h=csim('impuls',t,G); //Obtiene la respuesta al impulso
r=t; //Genera la entrada rampa
L=length(t); //obtiene la longitud de la respuesta
y=dt*conv(h,r); //Evalúa la integral de convolución (expresión (9.1))
y2=y(1,L); //Recorta tomando solo la duración de la respuesta
plot(t,y2,'r'); //Grafica respuesta en color rojo
```

Tercer método. Usando la respuesta al impulso unitario de  $\frac{G(s)}{s^2}$ :

```
G1=G/s^2; //función de transferencia del sistema G(s)/s^2
y3=csim('impuls',t,G1); //Obtiene la respuesta al impulso
plot(t,y3,'m:'); //Grafica respuesta en magenta punteado
```

Cuarto método. Usando la respuesta al escalón unitario de  $\frac{G(s)}{s}$ :

```
G2=G/s; //función de transferencia del sistema G(s)/s^2
y4=csim('step',t,G2); //Obtiene la respuesta al escalón
plot(t,y4,'c--'); //Grafica respuesta en cyan con guiones
legend('primer método','rampa unitaria','segundo método','tercer
método','cuarto método');
```




Figura 9.5.- Respuesta a la rampa unitaria del ejemplo.

Al hacer zoom en la figura 9.5 con ayuda de Scilab se puede observar que el primero, tercero y cuarto método producen respuestas prácticamente idénticas, sólo el método de la *convolución* produce resultados ligeramente distintos, ya que la *convolución* reemplaza al *solver* utilizado por *csim* para resolver las ecuaciones diferenciales implicadas en el sistema.

### Consideraciones teóricas:

Antes de utilizar Scilab para obtener la respuesta de un sistema ante una entrada de prueba es importante **saber que esperar**, ya que de esta manera podemos interpretar mejor el resultado proporcionado por Scilab, o bien, podemos detectar situaciones inesperadas o errores involuntarios al introducir los datos o configurar las escalas. Además el conocimiento teórico nos da elementos para decidir la utilidad de la simulación y extraer conclusiones para el análisis o el diseño.

Por esta razón se presenta la caracterización teórica con cierto detalle de la respuesta de sistemas de primer orden (en esta práctica) y segundo orden (en la siguiente práctica) y de ahí se infieren algunas predicciones para sistemas de orden mayor que 2.

### Sistemas de Primer Orden. Caracterización de la Respuesta al Escalón Unitario.

La función de transferencia de un sistema de primer orden siempre se puede escribir en la forma siguiente

$$G(s) = \frac{Y(s)}{X(s)} = \frac{K}{Ts + 1} \quad (9.6)$$

y está caracterizada por sus dos parámetros (positivos para el caso estable):

$K$  : Es la Ganancia del sistema (El valor de estado estable de la respuesta al escalón unitario)

$T$  : Es la constante de tiempo (Su valor caracteriza la duración del transitorio).

El **polo** correspondiente del sistema es

$$p = -1/T \quad (9.7)$$

La ecuación diferencial correspondiente a (9.6) para la entrada  $x(t)$  y la salida  $y(t)$  es

$$T \dot{y}(t) + y(t) = K x(t) \quad (9.8)$$

Resolviendo la ecuación diferencial anterior considerando una entrada escalón unitario y con la condición inicial  $y(0) = y_0$  se obtiene la siguiente expresión analítica exacta para la respuesta al escalón unitario

$$y(t) = y_0 e^{-t/T} + K(1 - e^{-t/T}) \quad (9.9)$$

En la siguiente tabla se obtienen algunos puntos de la respuesta típica dada por (9.9) para  $y_0 = 0$ :

| t        | y(t)    |
|----------|---------|
| 0        | 0       |
| T        | 0.6321K |
| 2T       | 0.8647K |
| 3T       | 0.9502K |
| 4T       | 0.9817K |
| 5T       | 0.9933K |
| 6T       | 0.9975K |
| ...      | ...     |
| $\infty$ | K       |

Por lo tanto se verifica que el valor de la respuesta cuando ha transcurrido mucho tiempo (valor final o de estado estable) es  $y(\infty) = K$ .

 Obsérvese también en la tabla anterior que cuando han transcurrido 5 constantes de tiempo

(5T), el transitorio del sistema prácticamente ha desaparecido, ya que la respuesta del sistema está al 99.33 % de su valor final.

La gráfica típica de la respuesta al escalón unitario de un sistema estable de primer orden siempre tiene la forma mostrada en la figura 9.6.




Figura 9.6.- Respuesta al escalón unitario de un sistema de primer orden.

**Ejemplo:** ¿Cuánto valen la ganancia, la constante de tiempo y el polo del sistema dado por la siguiente función de transferencia?

$$G(s) = \frac{4}{s+2} \quad (9.10)$$

**Solución:** Dividiendo numerador y denominador de la función de transferencia dada entre 2, se puede escribir como sigue

$$G(s) = \frac{2}{0.5s+1}$$

Comparando con la forma general (9.6), se obtiene:  $K = 2$ ,  $T = 0.5$ , por lo tanto,  $p = -2$ .

**Ejemplo:** Para el sistema de primer orden (9.10) del ejemplo anterior, elegir un intervalo de tiempo adecuado para la simulación y graficar mediante Scilab la respuesta al escalón unitario.

**Solución:** Como en la constante de tiempo es  $T = 0.5$  seg, el transitorio del sistema un poco más de  $5T = 2.5$  seg, por lo tanto un intervalo de tiempo de simulación adecuado será de unas 7 u 8 constantes de tiempo, eligiendo 7 constantes:  $7T = 3.5$  seg. Este intervalo nos permitirá apreciar completamente el transitorio y un poco del estado estable. Por lo tanto el código de Scilab queda

```
s=poly([0 1], 's', 'c'); //Variable de Laplace
G=4/(s+2); //función de transferencia del sistema
t=0:0.01:3.5; //Intervalo de tiempo de simulación
y=csim('step',t,G); //Obtiene la respuesta al escalón unitario
plot(t,y,'lineWidth',4); xgrid;
title('Respuesta al escalón unitario','fontSize',5);
```

```

xlabel('tiempo (seg)', 'fontSize', 4);
ylabel('Salida', 'fontSize', 4);
a=get('current_axes');
a.font_size=4;

```




Figura 9.7.- Respuesta al escalón unitario del sistema (9.10) del ejemplo.

### Respuesta al Impulso unitario de un sistema de Primer orden.

Si la entrada del sistema es un impulso unitario, entonces su respuesta en el dominio de Laplace es simplemente su Función de transferencia, es decir, como

$$\frac{Y(s)}{X(s)} = G(s) \quad (9.11)$$

Si la entrada es un impulso unitario, entonces  $X(s) = 1$  y por lo tanto

$$Y(s) = G(s) \quad (9.12)$$

es decir, la respuesta al impulso unitario en el dominio del tiempo será

$$y(t) = \mathcal{L}^{-1}[G(s)] \quad (9.13)$$

Y para el caso de primer orden

$$y(t) = \mathcal{L}^{-1}\left[\frac{K}{Ts+1}\right] \quad (9.14)$$

de donde se obtiene

$$y(t) = \frac{K}{T} e^{-t/T} \quad (9.15)$$

**Respuesta (Sin Entrada) a Condiciones Iniciales distintas de cero.**

Considerando la ecuación diferencial (9.8) con entrada cero se puede obtener la respuesta del sistema de primer orden ante condiciones iniciales distintas de cero  $y(0) = y_0$ ,

$$T \dot{y}(t) + y(t) = 0 \quad (9.16)$$

aplicando transformada de Laplace a ambos lados

$$T[sY(s) - y_0] + Y(s) = 0 \quad (9.17)$$

Despejando

$$Y(s) = \frac{T y_0}{Ts + 1} \quad (9.18)$$

Regresando al dominio del tiempo

$$y(t) = y_0 e^{-t/T} \quad (9.19)$$

**Ejemplo:** Obtener en Scilab la respuesta del sistema del sistema de primer orden (9.10) del ejemplo anterior con entrada cero, ante la condición inicial  $y(0) = 1$ .

**Solución 1.-** El modelo en función de transferencia no permite considerar condiciones iniciales diferentes de cero, para poder obtener la respuesta a condiciones iniciales no cero debemos obtener el modelo en la variable de estado  $x_1 = y$  a partir de la ecuación (9.8)

$$\dot{x}_1(t) = -\frac{1}{T} x_1(t) + \frac{K}{T} x(t) \quad (9.20)$$

(Donde  $x(t)$  hace el papel de la entrada). Es decir, en el formato estándar, las matrices A, b, c, d son escalares y son:

$$A = -\frac{1}{T}, \quad b = \frac{K}{T}, \quad c = 1, \quad d = 0 \quad (9.21)$$

Entonces, para obtener la respuesta con entrada cero a la condición inicial  $y(0) = 1$  en Scilab se puede hacer de la siguiente manera:

```
T=0.5; //Constante de Tiempo
K=2; //Ganancia del sistema
A=-1/T; b=K/T; c=1; d=0; //Matrices del modelo
Sist=syslin('c',A,b,c,d); //Modelo en variables de estado del sistema
t=0:0.01:3.5; //Intervalo de tiempo de simulación
x=zeros(t); //Entrada cero
y0=1; //Condicion inicial
y=csim(x,t,Sist,y0); //Obtiene la respuesta a la condicion inicial
plot(t,y,'lineWidth',4); xgrid;
title('Respuesta a la condicion inicial y(0)=1','fontSize',5);
xlabel('tiempo (seg)','fontSize',4);
ylabel('Salida','fontSize',4);
a=get('current_axes');
a.fontSize=4;
```

Obsérvese que el intervalo de tiempo adecuado para la simulación es el mismo que el usado para la respuesta al escalón. La gráfica de la respuesta obtenida se muestra en la figura 9.8.




Figura 9.8.- Respuesta del sistema (9.10) a la condición inicial  $y(0) = 1$

**Solución 2.-** Podemos utilizar el modelo en función de transferencia para obtener la respuesta a condiciones iniciales distintas de cero si comparamos la respuesta del sistema sin entrada ante la condición inicial  $y(0) = y_0$  dada por (9.19) con la respuesta al impulso unitario del sistema con condiciones iniciales cero dada por (9.15) se observa que ambas respuestas coinciden si el impulso se multiplica por  $(T/K)y_0$ .

En Scilab, usaremos el modelo en función de transferencia y obtendremos la respuesta al impulso unitario multiplicada por  $T/K$  (puesto que  $y_0 = 1$ ):

```
s=poly([0 1], 's', 'c'); //Variable de Laplace
G=4/(s+2); //función de transferencia del sistema
t=0:0.01:3.5; //Intervalo de tiempo de simulación
T=0.5; //Constante de Tiempo
K=2; //Ganancia del sistema
y=csim('impuls',t,G)*T/K; //Obtiene la respuesta al impulso por T/K
plot(t,y,'lineWidth',4); xgrid;
title('Respuesta a un impulso de amplitud K/T','fontSize',5);
xlabel('tiempo (seg)','fontSize',4);
ylabel('Salida','fontSize',4);
a=get("current_axes");
a.font_Size=4;
```

Obsérvese que el intervalo de tiempo adecuado para la simulación es el mismo que el usado para la respuesta al escalón. La gráfica obtenida se muestra en la figura 9.9. Compárese con la figura 9.8 y observe que son idénticas.




Figura 9.9.- Respuesta del sistema (9.10) a un impulso de amplitud T/K

#### Desarrollo de la Práctica.

1. Probar todos los ejemplos propuestos por el profesor conforme los va explicando.
2. Realizar todos los ejercicios propuestos.

#### Reportar:

1.- Mediante el Teorema del valor inicial, calcula el valor inicial  $y(0)$  de la respuesta al impulso unitario del sistema de primer orden (9.6). Compara el resultado obtenido con lo que predice la ecuación (9.15).

2.- Tomando en cuenta el resultado del problema anterior y recordando que el transitorio de un sistema de primer orden se extingue en aproximadamente 5 constantes de tiempo, Dibuja a mano la respuesta al impulso unitario aproximada del sistema dado por la función de transferencia siguiente ( $a, b$  son constantes positivas)

$$G(s) = \frac{b}{s + a}$$

3.- Para el circuito de la figura 9.10. Calcula la Ganancia y la constante de tiempo en términos de  $R$  y  $C$ . y Escribe los comandos de Scilab necesarios para obtener la respuesta al escalón unitario considerando  $R = 10\text{ K}\Omega$ ,  $C = 100\text{ }\mu\text{F}$  y la condición inicial  $v_C(0) = 1\text{ volt}$ . ¿Es lógico lo que se obtiene? explica.




Fig. 9.10. Circuito RC